

Principles of Linear Algebra With
Mathematica[®]
Rolling an Ellipse Along a Curve

Kenneth Shiskowski and Karl Frinkle

© Draft date March 12, 2012

Contents

1	Rolling an Ellipse Along a Curve	1
1.1	The Setup	1
1.2	The First Step	2
1.3	Automating the Process	8
1.4	Further Questions to Consider	13

Chapter 1

Rolling an Ellipse Along a Curve

1.1 The Setup

In Section 7.2 of Principles of Linear Algebra With *Mathematica*[®], we discussed the topic of “rolling” a circle along a curve. Of course, the rolling was in quotes due to the fact that we were really just sliding the circle along the curve. The animation have the appearance of a rolling circle, but this was due only to the fact a circle is a perfectly symmetric object, and no matter which way it is rotated, there is no apparent change in orientation. So what happens if we pick an object which is not as symmetric? Take, for instance, an ellipse. Clearly if an ellipse is rotated through most angles, it will be obvious to an observer that the ellipse was indeed rotated. The only two angles for which this is not the case is $\theta = \pi$ and $\theta = 2\pi$. The question then becomes: what additionally must be done to rotate an ellipse versus a circle?

To determine some of the math required to rotate an ellipse, we will start with a simple setup, that of rolling the ellipse along a horizontal line, similar to rolling a football, along a level surface, end over end. We define an ellipse with center at (x_c, y_c) parametrically as

$$\begin{aligned}x(t) &= x_c + a \cos(t) \\y(t) &= y_c + b \sin(t)\end{aligned}\tag{1.1}$$

where a and b are the length of the major/minor axes corresponding, dependent upon $a > b$ or $a < b$. If we graph this ellipse, starting at $t = 0$, then initially we have the point $(x_c + a, y_c)$. The bottom most point on the ellipse occurs when $t = \frac{3}{2}\pi$, yielding the point $(x_c, y_c - b)$. So in order to roll an ellipse along

a horizontal line, our line must have the equation

$$y = y_c - b \tag{1.2}$$

To perform actual calculations, we must choose values for the center (x_c, y_c) and lengths of axes, a and b , however it should be apparent throughout the following calculations that nothing depends on these values. We will choose our center to be at $(2, 4)$, with $a = 4$ and $b = 2$. Thus our horizontal line, given in equation (1.2), is $y = 4 - 2 = 2$. We will have *Mathematica* graph our initial setup now.

```
xc = 2; yc = 4;
a = 4; b = 2;
EllipseF[t_] = {xc + a Cos[t], yc + b Sin[t]};
EllipsePlot = ParametricPlot[Evaluate[EllipseF[t], {t, 0, 2π}], PlotStyle→{
Thickness[0.007], Red}];
LinePlot = Graphics[{Thickness[0.005], Pink, Line[{{-10, 2}, {20,
2}}]}];
CenterPlot = Graphics[{PointSize[0.015], Black, Point[{{xc, yc}}]}];
Show[Evaluate[EllipsePlot, CenterPlot, LinePlot], AxesOrigin→{0, 0}, Plot-
Range→{{-10, 20}, {0, 10}}]
```




Figure 1.1: The original ellipse and the horizontal line it will roll along.

As you can see from Figure 1.1, the ellipse rests on the horizontal line that it will roll along. So we have now set up our problem.

1.2 The First Step

For the next step in the process, we simply wish to roll our ellipse a fixed distance, r , to the right. If we wish to roll the ellipse r units in the x -direction,

then we must count out r units on the ellipse, in a counter-clockwise direction, starting at the point on the ellipse which is touching the line.

Recall that the formula for the arc length L of a parametric curve $(x(t), y(t))$ for $t \in (t_0, t_1)$ is given by

$$L = \int_{t_0}^{t_1} \sqrt{(x'(t))^2 + (y'(t))^2} dt \quad (1.3)$$

The total arclenth of the ellipse can be found by setting $t_0 = 0$ and $t_1 = 2\pi$, which gives a full revolution.

```
NIntegrate[Norm[D[EllipseF[t], t]], {t, 0, 2 π}]
```

```
19.3769
```

```
FindRoot[NIntegrate[Norm[D[EllipseF[t], t]], {t,  $\frac{3\pi}{2}$ , t1}] == 2,
{t1, 1.1  $\frac{3\pi}{2}$ ,  $\frac{3\pi}{2}$ , 2π}]
```

```
NIntegrate::nlim: t = t1 is not a valid limit of integration. >>
```

```
{t1 → 5.22924}
```

We plot the difference of 2 and the arclength formula, notice the root here corresponds to the t value which give arclength 2.

```
Plot[2 - NIntegrate[Norm[D[EllipseF[t], t]], {t,  $\frac{3\pi}{2}$ , t1}], {t1, 0, 2π}]
```




Figure 1.2: Difference of 2 and the arclength of the ellipse.

```
EllipseF[5.229238280827457]
```

```
{3.97657, 2.26124}
```

```
TouchPlots = Graphics[{{PointSize[0.025], Green, Point[{EllipseF[ $\frac{3\pi}{2}$ ],
EllipseF[5.229238280827457]}]}];
```

```
NextTouch = Graphics[{PointSize[0.025], Black, Point[{{4, 2}}]};
Show[EllipsePlot, TouchPlots, CenterPlot, LinePlot, NextTouch,
AxesOrigin→{0, 0}, PlotRange→{{-3, 8}, {0, 7}}]
```




Figure 1.3: Original point on the ellipse which touches the line, and the new point which will touch the line after rolling two units to the right.

```
Show[EllipsePlot, TouchPlots, CenterPlot, LinePlot, NextTouch,
AxesOrigin→{1, 1}, PlotRange→{{1, 5}, {1, 3}}]
```




Figure 1.4: Close-up of Figure 1.3 near the desired point of rotation.

Now we need to rotate the ellipse. But we are missing one important piece of information, the angle of rotation. We need to construct the pair of vectors which will yield the correct angle of rotation. As a first guess, you may try to use the three points depicted in Figure 1.4 above, but performing the resulting calculations show that this will not work. The next obvious idea is to still use the horizontal line for one vector, but as second vector, use the tangent line at the new point on the ellipse which will be touching the horizontal line after

rotation. This also makes us determine where this tangent line intercepts the horizontal line $y = 2$. We illustrate this in the *Mathematica* code and Figure 1.5 following.

```
M = D[EllipseF[t], t][[2]]/D[EllipseF[t], t][[1]]
```

```
- Cot[t]
-----
      2
```

```
m = M /. t -> 5.229238280827457
```

```
0.284193
```

```
Solve[(y - EllipseF[5.229238280827457][[2]] == m (x -
EllipseF[5.229238280827457][[1]])] /. y -> 2]
```

```
{{x -> 3.05734}}
```

```
AnglePoint = Graphics[{PointSize[0.025], Blue,
Point[{{3.0573416546670975, 2}}]}];
```

```
ArrowPlots = Graphics[{Arrowheads[.10], Thickness[.010], Black,
Arrow[{{3.0573416546670975, 2}, {3.976573754490907,
2.2612392205181213}}], Black, Arrow[{{3.0573416546670975, 2}, {4,
2}}]}];
```

```
Show[EllipsePlot, CenterPlot, LinePlot, NextTouch, ArrowPlots,
TouchPlots, AnglePoint, PlotRange -> {{2.9, 4.2}, {1.8, 2.4}}, Axes-
Origin -> {2.9, 1.8}]
```




Figure 1.5: The vectors needed to compute the angle of rotation for the first step in the rolling.

```
CenCirc = {3.0573416546670975, 2}
```

```
{3.05734, 2}
```

```
TopCirc = EllipseF[5.229238280827457]
```

```
{3.97657, 2.26124}
```

```
BotCirc = {4, 2}
```

```
{4, 2}
```

```
U = TopCirc - CenCirc
```

```
{0.919232, 0.261239}
```

```
V = BotCirc - CenCirc
```

```
{0.942658, 0}
```

```
thetaVal = VectorAngle[U, V]
```

```
0.276893
```

```
thetaVal  $\frac{180}{\pi}$ 
```

```
15.8648
```

So from above, we have approximately 15.86 degrees to rotate, or 0.28 radians. So now we need to move the ellipse to the origin, rotate, and then finally place it back so that its center is back at the original center. To do this, we take the ellipse function $F(t)$, and subtract the center $(2, 4)$ from the corresponding components. Next, we rotate using the rotation matrix

$$A_\theta = \begin{bmatrix} \cos(\theta) & -\sin(\theta) \\ \sin(\theta) & \cos(\theta) \end{bmatrix} \quad (1.4)$$

and our formula for the rotated ellipse, $F_\theta(t)$ is now given by

$$F_\theta(t) = A_\theta(F(t) - (2, 4)) + (2, 4). \quad (1.5)$$

```
EllipseShift = EllipseF[t] - {2, 4}
```

```
{4Cos[t], 2Sin[t]}
```

```
RotMatrix[theta_] = {{Cos[theta], -Sin[theta]}, {Sin[theta], Cos[theta]}};
```

```
(NewEllipse[t_] = RotMatrix[-thetaVal].EllipseShift) // MatrixForm
```

```
 $\begin{pmatrix} 3.84764 \text{Cos}[t] + 0.546736 \text{Sin}[t] \\ -1.09347 \text{Cos}[t] + 1.92382 \text{Sin}[t] \end{pmatrix}$ 
```

```
NewEllipsePlot = ParametricPlot[Flatten[NewEllipse[t] + {2, 4}],  
{t, 0, 2 Pi}, PlotStyle -> {Thickness[0.007], Blue}];
```

```
Show[NewEllipsePlot, EllipsePlot, TouchPlots, CenterPlot, LinePlot,
```

`NextTouch, AxesOrigin→{0, 0}, PlotRange→{{-3, 8}, {0, 7}}`




Figure 1.6: Original ellipse and the rotated ellipse, both at center (2, 4).

We have successfully rotated the ellipse, now we need to move it into the correct position. This must be a simple shift of some kind, but we need to find the right one, keeping in mind that we wish to automate this process. The one thing we must be sure of, is that the lowest point on the rotated ellipse (corresponding to $t = 5.229238282$) must move to the point (4, 2) since that is how far to the right we are rolling the ellipse along the horizontal line. Correspondingly, the original center (2, 4) must therefore be shifted the same amount!

`LowestPoint = NewEllipse[5.229238280827457]`

`{1.42596, -2.21286}`

`ShifT = BotCirc - LowestPoint`

`{2.57404, 4.21286}`

`(NewNewEllipse[t_] = NewEllipse[t] + ShifT) // MatrixForm`

$$\begin{pmatrix} 2.57404 + 3.84764 \cos[t] + 0.546736 \sin[t] \\ 4.21286 - 1.09347 \cos[t] + 1.92382 \sin[t] \end{pmatrix}$$

`NewNewEllipsePlot = ParametricPlot[Flatten[NewNewEllipse[t]],
{t, 0, 2 π}, PlotStyle→{Thickness[0.007], Blue}];`

`NewCenterPlot = Graphics[{PointSize[0.015], Black,
Point[{ShifT}]}];`

`Show[NewNewEllipsePlot, EllipsePlot, TouchPlots, CenterPlot,
LinePlot, NextTouch, NewCenterPlot, AxesOrigin→{0, 0},`

PlotRange $\rightarrow \{\{-3, 8\}, \{0, 7\}\}$




Figure 1.7: Original ellipse and the rotated ellipse, both in the correct positions.

1.3 Automating the Process

Now that we know how an ellipse can be rolled to to right a single fixed distance to the right, we need to determine how to create a process for repeating the process. We will start the process from the previous section over, this time making things more flexible so that the process can be repeated in a **For** loop.

The variables **xc** and **yc** will correspond to the center of the ellipse, and **a** and **b** are the lengths along the *x*-axis and *y*-axis of the ellipse, respectively.

xc = 2; yc = 4;

a = 4; b = 2;

The definition of the ellipse, as a function of *t*, where $0 \leq t \leq 2\pi$ is given next.

EllipseF[t_] = {xc + a Cos[t], yc + b Sin[t]};

The variable **r** corresponds to how far to the right we wish to go for each step. The variable **yline** is the *y* =line that the ellipse will roll along.

r = 0.5;

yline = yc - b;

Next we calculate the total perimeter of the ellipse. This will help is later, when we need to use the **FindRoot** command to locate the times that sweep

out arclengths of length r given above, which is where we come up with the formula in the definition of the variable S .

```
TotalPerimeter = NIntegrate[Norm[D[EllipseF[t], t]], {t, 0, 2 π}]
```

```
19.3769
```

The variable **NumRSteps** tells us how many steps to the right we will go, $A[\theta]$ is the standard rotation matrix, $t0$ is the initial time corresponding to the point on the ellipse that touches the line **yline**. The variable **xval** is where we start the ellipse rolling, in the x -direction.

```
NumRSteps = 40;
```

```
S = (TotalPerimeter/r)/NumRSteps
```

```
0.968845
```

```
A[θ_] = {{Cos[θ], -Sin[θ]}, {Sin[θ], Cos[θ]}};
```

$$\begin{pmatrix} \cos[\theta] & -\sin[\theta] \\ \sin[\theta] & \cos[\theta] \end{pmatrix}$$

```
t0 = N[ $\frac{3\pi}{2}$ ];
```

```
xval = xc;
```

We will want to plot each ellipse, and corresponding center, as we go through the **For** loop. Therefore we initialize a couple of plots.

```
PlotEllipse[0] = ParametricPlot[EllipseF[t], {t, 0, 2 π}, PlotStyle→  
{Thickness[0.007], Blue}];
```

```
PlotCenter[0] = Graphics[{PointSize[0.015], Black, Point[{xc,  
yc}]}];
```

```
LinePlot = Graphics[{Thickness[0.005], Pink, Line[{{-10, 2}, {30,  
2}]}];
```

We now do the **For** loop which, which automates each step from the previous section. See if you can determine exactly how the following code works, and relate it back to the steps used in Section 1.2.

```
For[k = 1, k ≤ NumRSteps, k ++,  
  ts = t1 /. FindRoot[NIntegrate[Norm[D[EllipseF[t], t]], {t, t0, t1}]  
  == r, {t1, t0, t0, t0 + 3 S}];  
  M = D[EllipseF[t][[2]], t]/D[EllipseF[t][[1]], t];
```

```

m = M /. {t -> ts};
xint = (x /. Solve[(y - EllipseF[ts][[2]] == m (x - EllipseF[ts][[1]]))
/. y -> yline, x][[1]]);
Cpt = {xint, yline};
Tpt = EllipseF[ts];
Bpt = {xval + k*r, yline};
u = Tpt - Cpt;
v = Bpt - Cpt;
theta = VectorAngle[u, v];
EllipseCol[t_] = EllipseF[t] - {xc, yc};
NewEllipse[t_] = Simplify[A[-theta].EllipseCol[t]];
Lpt = NewEllipse[ts];
Shift = Bpt - Lpt;
EllipseF[t_] = NewEllipse[t] + Shift;
xc = Shift[[1]];
yc = Shift[[2]];
PlotEllipse[k] = ParametricPlot[EllipseF[t], {t, 0, 2 pi}, PlotStyle ->
{Thickness[0.007], Blue}];
PlotCenter[k] = Graphics[{PointSize[0.015], Black, Point[{xc,
yc}]}];
Print[{k, t0, ts, m, xint, -theta}];
t0 = ts;
]
NIntegrate::nlim: t = t1 is not a valid limit of integration. >>
{1, 4.71239, 4.83763, 0.0629521, 2.25082, -0.062869}
NIntegrate::nlim: t = t1 is not a valid limit of integration. >>
{2, 4.83763, 4.96438, 0.0652499, 2.75234, -0.0651575}
NIntegrate::nlim: t = t1 is not a valid limit of integration. >>
General::stop: Further output of NIntegrate::nlim will be suppressed during
this calculation. >>
{3, 4.96438, 5.09426, 0.0702456, 3.25399, -0.0701304}
{4, 5.09426, 5.22924, 0.0788985, 3.7559, -0.0787354}
{5, 5.22924, 5.37184, 0.0931575, 4.25821, -0.0928894}
{6, 5.37184, 5.52556, 0.116996, 4.76112, -0.116466}
{7, 5.52556, 5.69552, 0.158918, 5.26484, -0.1576}
{8, 5.69552, 5.88903, 0.237402, 5.76913, -0.233087}
{9, 5.88903, 6.11347, 0.380616, 6.2707, -0.363685}

```

{10, 6.11347, 6.36084, 0.526421, 6.75906, -0.48456}
{11, 6.36084, 6.59657, 0.447388, 7.23896, -0.42068}
{12, 6.59657, 6.80115, 0.282734, 7.73273, -0.275542}
{13, 6.80115, 6.979, 0.18273, 8.23491, -0.180736}
{14, 6.979, 7.13813, 0.129972, 8.73828, -0.129248}
{15, 7.13813, 7.28444, 0.100698, 9.2412, -0.10036}
{16, 7.28444, 7.42196, 0.083461, 9.74358, -0.0832681}
{17, 7.42196, 7.55351, 0.0730009, 10.2455, -0.0728716}
{18, 7.55351, 7.68123, 0.0667793, 10.7472, -0.0666803}
{19, 7.68123, 7.80686, 0.0635364, 11.2488, -0.0634511}
{20, 7.80686, 7.93193, 0.0626904, 11.7503, -0.0626085}
{21, 7.93193, 8.05793, 0.0640984, 12.2518, -0.0640108}
{22, 8.05793, 8.18642, 0.0680005, 12.7533, -0.0678959}
{23, 8.18642, 8.31922, 0.0751135, 13.2551, -0.0749727}
{24, 8.31922, 8.4586, 0.0869358, 13.7573, -0.0867178}
{25, 8.4586, 8.60763, 0.106491, 14.2599, -0.106091}
{26, 8.60763, 8.77073, 0.140125, 14.7633, -0.139218}
{27, 8.77073, 8.95437, 0.201724, 15.2675, -0.199053}
{28, 8.95437, 9.16669, 0.318175, 15.7709, -0.308047}
{29, 9.16669, 9.40844, 0.486882, 16.2656, -0.453098}
{30, 9.40844, 9.65271, 0.50424, 16.7456, -0.467034}
{31, 9.65271, 9.86923, 0.338826, 17.2334, -0.326686}
{32, 9.86923, 10.0562, 0.213117, 17.7337, -0.209976}
{33, 10.0562, 10.2215, 0.146155, 18.237, -0.145127}
{34, 10.2215, 10.3721, 0.109891, 18.7402, -0.109452}
{35, 10.3721, 10.5126, 0.088961, 19.2427, -0.0887274}
{36, 10.5126, 10.6461, 0.0763462, 19.7448, -0.0761984}
{37, 10.6461, 10.775, 0.0687253, 20.2466, -0.0686175}
{38, 10.775, 10.9012, 0.0644568, 20.7482, -0.0643678}
{39, 10.9012, 11.0264, 0.0627452, 21.2497, -0.062663}

```
{40, 11.0264, 11.1518, 0.0632966, 21.7512, -0.0632122}
```

Notice that for each step in the **For** loop, the values of the variables **k**, **t0**, **ts**, **m**, **xint**, and $-\theta_{\text{val}}$ are displayed. As a general rule, this is an important tool in helping to determine the behavior (or lack thereof) of your code as it progresses through successive iterations of the loop. Now that we have completed the rotation, we simply need to put the plots together in a **Manipulate** command.

```
Manipulate[Show[PlotEllipse[h], PlotCenter[h], LinePlot, PlotRange
→{{-5, 30}, {0, 10}}, AxesOrigin→{0, 0}, ImageSize→{400, 200}],
{{h, 0, "Step"}, 0, 40, 1}]
```




Figure 1.8: The 27th frame in the animation of rolling the ellipse.

```
Show[Table[PlotCenter[k], {k, 0, 40}], PlotRange→{{-5, 30}, {0,
10}}, Axes→True, AxesOrigin→{0, 0}]
```



Figure 1.9: The centers of each of the ellipses throughout the rotation.

Figure 1.8 is the culmination of all of our work thus far. Unfortunately, we cannot display the actual animation here, but it appears that we have successfully rolled an ellipse along a horizontal line! Showing all 41 ellipses at the same time would not help us view the process at all. In Figure 1.9, the center of each ellipse is displayed. Note that the distance between centers is significantly less when the curvature of the portion of the ellipse resting on the line is smallest. This corresponds to the portions of the ellipse closest to the minor axes.

1.4 Further Questions to Consider

We end with a discussion (and leave you with questions) on how to generalize this process. Before we do this, we must realize what restrictions were used in all the work done previous. The most important assumption that was made, was that our curve that we rolled the ellipse along was actually a horizontal line.

1. How do things change if we decide not to use a horizontal line, but a line with nonzero slope instead?

If you can determine the answer to this question, then the following should be answerable as well. Remember, that if we were rolling an ellipse a distance r to the right along a horizontal line, where the ellipse touched the line at the point (x_p, y_p) , we then knew that the rotated ellipse would be touching the horizontal line at the point $(x_p + r, y_p)$. This is clearly not the case for a line with nonzero slope! Furthermore, we also use the fact that the line was horizontal to help compute the angle of rotation needed. This must change as well! If you can figure out how to adapt the previous work to allow for lines on nonzero slope, then the following questions may also be answerable.

2. What must be modified to roll an ellipse along a curve given in the form $y = f(x)$?

3. What happens if you wish to roll an ellipse along a parametric curve of the form $(x, y) = (x(s), y(s))$? Is this really a generalization of question 1 or is more mathematics required?